
# This query implements the geologic symbols for
# strike/dip of bedding, including special cases of
# horizontal, vertical and overturned beddings

LineStyleDropAnchor(0) # center of symbol
LineStyleSetColor(0,0,0) #black symbols
LineStyleSetCapJoinType(1,1) #squared off lines

strike1 = bedding.strike
dip1 = bedding.dip
scaling = 7
radius = 4*scaling
LineStyleSetLineWidth(scaling)
textheight = 6 *scaling

#Draw Strike Line
if(dip1!=0){
  LineStyleLineTo(strike1,16*scaling)
  LineStyleMoveToAnchor(0)
  LineStyleLineTo(strike1 - 180,16*scaling)
  LineStyleMoveToAnchor(0)
}
#rotate symbol by 180 when dip direction indicates
if (bedding.dip_direction)  rot = 180  else rot = 0
markangle = strike1-90+rot
if (markangle > 360) {markangle = markangle -360}
if (markangle < 0)  {markangle = markangle + 360}
if (bedding.overturned){  #add arc to indicate overturned bedding
    LineStyleDrawArc(strike1,radius,radius,radius,strike1+rot,180,0,0,0)
    LineStyleLineTo(markangle,4*scaling)
    }
else {
 if (dip1==90){ # use a cross bar to mark vertical bedding
        LineStyleDrawRectangle(6*scaling,scaling,markangle,1)
  }
  else {
   if (dip1==0){ # horizontal bedding -circle with cross inset
       LineStyleDrawRectangle(4*radius,scaling,0,1)
       LineStyleDrawCircle(2*radius,0)
       LineStyleDrawRectangle(4*radius,scaling,90,1)
    }
    else { # dip direction line
     LineStyleLineTo(markangle,4*scaling)
    }
   }
 }
if ((dip1<90) AND (dip1>0)){ #label non-horizontal and non-vertical points
    LineStyleSetTextColor(0,0,0)
    str$ = sprintf(“%2d”,dip1);
    if ((markangle>90) and (markangle<270)){
       LineStyleTextNextPosition(str$,textheight,0,0,nextx,nexty,tlength)
        LineStyleMoveTo(180,tlength)
        }
    if (markangle>180){
        LineStyleMoveTo(- 90,textheight)
        }
    LineStyleDrawText(str$,textheight,0,2)
}

Points: Geologic Bedding Dip and Strike
Query generates a bedding symbol based on dip, strike and dip direction. This
query is data dependent, change bedding.strike, bedding.dip and
bedding.dip_direction  to values appropriate for your data.


